

Día 4 -Reproductor de música

Lo primero es crear un archivo ActionScript 2.0. Vamos a crear cinco botones para cada una de las canciones a ser reproducidas. Además debemos hacer los botones de Stop, Play, Pausa, Aumento y disminución de volumen, y asignar a cada uno de ellos un nombre de instancia.

Los botones de música deben llamarse todos iguales y solo diferenciarse por el número final. Esto nos ahorrará mucha programación.

En el cuadro 1 de la Scene 1 escribimos el siguiente código

```
//---Crear la variable sonido común
var sonido:Sound = new Sound();
//---Acciones de los botones
var cantidad:Number = 5;
var pos:Number;
var volumen:Number = 100;
for(var i:Number = 1; i <= cantidad; i++){
    this["cancion" + i].numero = i;
    this["cancion" + i].onPress = function():Void{
        sonido.stop();
        pos = 0;
        sonido.loadSound("0" + this.numero + ".mp3", true);
    }
}
//---Acciones de los botones de reproducción
this.musplay.onPress = function(){
    sonido.stop();
    sonido.start(pos / 1000, 0);
}
this.muspause.onPress = function(){
    pos = sonido.position;
    sonido.stop();
}
this.musstop.onPress = function (){
    pos = 0
    sonido.stop();
}
//---Acciones de los botones de volumen
this.baja.onPress = function(){
    if (volumen > 0){
        volumen-=10;
        sonido.setVolume(volumen);
    }
}
this.sube.onPress = function(){
    if (volumen < 100){
        volumen+=10;
        sonido.setVolume(volumen);
    }
}
```

```
//---Slider de control de volumen
slider.slideBar._y = -50;
slider.slideBar.onEnterFrame = function() {
    sonido.setVolume(0-this._y);
};
slider.slideBar.onPress = function() {
    startDrag(this, false, this._x, -100, this._x, 0);
};
slider.slideBar.onRelease = slider.slideBar.onReleaseOutside=function () {
    stopDrag();
};
```